

3-D Data Plotting with DISLIN

1. DISLIN web site is at www.dislin.de. The web site has an online manual and example plots (along with sample Fortran 90 code) that are very useful.
2. We are currently using DISLIN with the Intel compiler. To compile, link, and run a program using the DISLIN graphics library, use the command

```
gf95link -a -r8 source-file-name
```

where *source-file-name* is given without the .f90 ending.
3. You must USE the dislin fortran module DISLIN found in the file `dislin.f90`. This file can be copied into your working directory from `/usr/local/dislin/gf/real64`. The use `dislin` statement should be placed in any program component (main program or subprogram) that calls a DISLIN routine. The module must be compiled (with the command `gfortran -c dislin.f90`) before using it in any program.
4. All floating point arguments to DISLIN subroutines **must** be of `kind=selected_real_kind(15)`.
5. All character strings passed to DISLIN as control parameters can be either upper or lower case.
6. The simple 3-D plot shown on the next page was generated with the following program:

```
program dislin3dplot
  use dislin
  implicit none
  integer, parameter::dp=selected_real_kind(15)
  integer::i,j
  integer::nx,ny !declared size of x and y array (and rows and columns of z array)
  real(dp)::xa=0.0_dp,xe=3000.0_dp,xor=0.0_dp,xstep=500.0_dp,ya=0.0_dp,ye=3000.0_dp,&
    yor=0.0_dp,ystep=500.0_dp,za=40.0_dp,ze=50.0_dp,zor=40.0_dp,zstep=2.0_dp
  real(dp), dimension(:),allocatable::x,y !array of x and y values where z known
  real(dp), dimension(:,:),allocatable::z

  !Sample program to plot a 3-d mesh data plot

  open(11,file="gwwlab.out")
  !Read in or compute nx and ny
  read(11,*)nx,ny
  allocate(x(nx),y(ny),z(nx,ny))
  !Compute or read in x and y values where z will be specified
  do i=1,nx
    x(i)=xa+dbble(i-1)*100.0_dp
  end do
  do i=1,ny
    y(i)=ya+dbble(i-1)*100.0_dp
  end do
  !Read in the z values from the file a row at a time
  do i=1,nx
    read(11,*)(z(i,j),j=1,ny)
  end do
  !
  call metafl("xwin") ! "XWIN" or "PS", "EPS", "PDF", "WMF" "BMP"
  call setpag("USAL") !"USAL" is US size A landscape, "USAP" is portrait
  call sclmod("FULL") !Scale the graphics window to fill the page
  call scrmod("REVERS") !sets black on white background
  call disini
  call complx ! Sets the font
  call name("Distance (m)","X")
  call name("Distance (m)","Y")
  call name("Head (m)","Z")
  call graf3d(xa, xe, xor, xstep, ya, ye, yor, ystep, za, ze, zor, zstep)
    ! xa, xe are the lower and upper limits of the x-axis.
    ! xor, xstep are the first x-axis label and the step between labels.
    ! ya, ye are the lower and upper limits of the y-axis.
    ! yor, ystep are the first z-axis label and the step between labels.
    ! za, ze are the lower and upper limits of the z-axis.
    ! zor, zstep are the first z-axis label and the step between labels.
  ! Use one of the following
  call surmat(z,nx,ny,1,1) !plots a mesh
  ! or
  !call surshd(x,nx,y,ny,z) !plots a color shaded surface
  call disfin ! finishes the plot
  stop
end program dislin3dplot
```

